pragma solidity ^0.4.21;

// File: zeppelin/contracts/math/SafeMath.sol

/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {
  function mul(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a * b;
    assert(a == 0 || c / a == b);
    return c;
  }

  function div(uint256 a, uint256 b) internal constant returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  function sub(uint256 a, uint256 b) internal constant returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  function add(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }
}

// File: zeppelin/contracts/ownership/Ownable.sol

/**
 * @title Ownable
 * @dev The Ownable contract has an owner address, and provides basic authorization control
 * functions, this simplifies the implementation of "user permissions".
 */
contract Ownable {
  address public owner;


  event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);


  /**
   * @dev The Ownable constructor sets the original `owner` of the contract to the sender
   * account.
   */
  function Ownable() {
    owner = msg.sender;
  }


  /**
   * @dev Throws if called by any account other than the owner.
   */
  modifier onlyOwner() {
    require(msg.sender == owner);
    _;
  }


  /**
   * @dev Allows the current owner to transfer control of the contract to a newOwner.
   * @param newOwner The address to transfer ownership to.
   */
  function transferOwnership(address newOwner) onlyOwner public {
    require(newOwner != address(0));
    OwnershipTransferred(owner, newOwner);
    owner = newOwner;
  }

}

// File: zeppelin/contracts/token/ERC20Basic.sol

/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  uint256 public totalSupply;
  function balanceOf(address who) public constant returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

// File: zeppelin/contracts/token/BasicToken.sol

/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));

    // SafeMath.sub will throw if there is not enough balance.
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public constant returns (uint256 balance) {
    return balances[_owner];
  }

}

// File: zeppelin/contracts/token/ERC20.sol

/**
 * @title ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/20
 */
contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public constant returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: zeppelin/contracts/token/StandardToken.sol

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));

    uint256 _allowance = allowed[_from][msg.sender];

    // Check is not needed because sub(_allowance, _value) will already throw if this condition is not met
    // require (_value <= _allowance);

    balances[_from] = balances[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    allowed[_from][msg.sender] = _allowance.sub(_value);
    Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(address _owner, address _spender) public constant returns (uint256 remaining) {
    return allowed[_owner][_spender];
  }

  /**
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   */
  function increaseApproval (address _spender, uint _addedValue)
    returns (bool success) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  function decreaseApproval (address _spender, uint _subtractedValue)
    returns (bool success) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}

// File: contracts/EpigenCareCrowdsale.sol

contract EpigenCareCrowdsale is Ownable {
  using SafeMath for uint256;

  StandardToken public token;

  uint256 public startTime;
  uint256 public endTime;
  address public wallet;
  address public tokenPool;
  uint256 public rate;
  uint256 public weiRaised;
  uint256 public weiPending;
  uint256 public tokensPending;
  uint256 public minimumInvestment;

  mapping (address => Transaction) transactions;
  mapping (address => bool) approvedAddresses;
  mapping (address => bool) verifiers;

  struct Transaction { uint weiAmount; uint tokenAmount; }

  event TokenPurchaseRequest(address indexed purchaser, address indexed beneficiary, uint256 value);

  function EpigenCareCrowdsale(uint256 _startTime, uint256 _endTime, uint256 _rate, address _wallet, address _tokenPool, address _token) Ownable() {
    require(_endTime >= _startTime);
    require(_rate > 0);
    require(_wallet != 0x0);
    require(_tokenPool != 0x0);

    token = StandardToken(_token);
    startTime = _startTime;
    endTime = _endTime;
    wallet = _wallet;
    tokenPool = _tokenPool;

    verifiers[msg.sender] = true;
    rate = _rate;
    minimumInvestment = 0.5 ether;
  }

  function () payable {
    requestTokens(msg.sender);
  }

  function requestTokens(address beneficiary) sufficientApproval(msg.value) public payable {
    require(beneficiary != 0x0);
    require(validPurchase());
    require(msg.value >= minimumInvestment);

    uint256 weiAmount = msg.value;
    uint256 tokens = weiAmount.mul(rate);

    if(approvedAddresses[beneficiary]) {
      weiRaised = weiRaised.add(weiAmount);

      token.transferFrom(tokenPool, beneficiary, tokens);
      wallet.transfer(weiAmount);
    } else {
      Transaction transaction = transactions[beneficiary];
      transaction.weiAmount = transaction.weiAmount.add(weiAmount);
      transaction.tokenAmount = transaction.tokenAmount.add(tokens);

      weiPending = weiPending.add(weiAmount);
      tokensPending = tokensPending.add(tokens);
      TokenPurchaseRequest(msg.sender, beneficiary, weiAmount);
    }
  }

  function validateTransaction(address purchaser) onlyVerifiers(msg.sender) {
    Transaction transaction = transactions[purchaser];

    weiRaised = weiRaised.add(transaction.weiAmount);
    weiPending = weiPending.sub(transaction.weiAmount);
    tokensPending = tokensPending.sub(transaction.tokenAmount);
    approvedAddresses[purchaser] = true;

    token.transferFrom(tokenPool, purchaser, transaction.tokenAmount);
    wallet.transfer(transaction.weiAmount);
    transaction.weiAmount = 0;
    transaction.tokenAmount = 0;
  }

  function pendingTransaction(address user) returns (uint value){
    return transactions[user].weiAmount;
  }

  function revokeRequest() {
    Transaction transaction = transactions[msg.sender];
    weiPending = weiPending.sub(transaction.weiAmount);
    tokensPending = tokensPending.sub(transaction.tokenAmount);
    msg.sender.transfer(transaction.weiAmount);
    transaction.weiAmount = 0;
    transaction.tokenAmount = 0;
  }

  modifier sufficientApproval(uint value) {
    uint tokensNeeded = tokensPending.add(value.mul(rate));
    uint tokensAvailable = token.allowance(tokenPool, this);
    require(tokensAvailable >= tokensNeeded);
    _;
  }

  function rejectRequest(address user, uint fee) onlyVerifiers(msg.sender) {
    Transaction transaction = transactions[user];
    weiPending = weiPending.sub(transaction.weiAmount);
    tokensPending = tokensPending.sub(transaction.tokenAmount);
    if(fee > 0) {
      transaction.weiAmount = transaction.weiAmount.sub(fee);
      wallet.transfer(fee);
    }

    user.transfer(transaction.weiAmount);
    transaction.weiAmount = 0;
    transaction.tokenAmount = 0;
  }

  function validPurchase() internal constant returns (bool) {
    bool withinPeriod = (now >= startTime && now <= endTime);
    bool nonZeroPurchase = msg.value != 0;
    return (withinPeriod && nonZeroPurchase);
  }

  function hasEnded() public constant returns (bool) {
    return now > endTime;
  }

  function updateMinimumInvestment(uint _minimumInvestment) onlyOwner {
    minimumInvestment = _minimumInvestment;
  }

  function updateRate(uint _rate) onlyOwner {
    rate = _rate;
  }

  function setVerifier(address verifier, bool value) onlyOwner {
    verifiers[verifier] = value;
  }

  function isValidated(address user) returns (bool) {
    return approvedAddresses[user];
  }

  modifier onlyVerifiers(address sender) {
    require(verifiers[sender]);
    _;
  }
}